# $Id$
# Maintainer: Tobias Powalowski <tpowa@archlinux.org>
# Contributor: Surain Mallawa Arachchi <mas@crc32.net>

pkgname=linux-atm
pkgver=2.5.2
pkgrel=4
pkgdesc="Drivers and tools to support ATM networking under Linux."
arch=('i686' 'x86_64')
depends=('glibc')
url="http://linux-atm.sourceforge.net/"
license=(GPL)
source=(http://downloads.sourceforge.net/sourceforge/${pkgname}/${pkgname}-${pkgver}.tar.gz
        man-pages.patch)
options=(!makeflags)
md5sums=('d49499368c3cf15f73a05d9bce8824a8'
         '181390736336cdb615ae32f1ae5acfa6')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  patch -Np1 -i ../man-pages.patch

  # install firmware in /usr/lib
  sed -i "s#/lib/firmware#/usr/lib/firmware#" src/extra/Makefile.in

  ./configure --prefix=/usr --sbindir=/usr/bin --sysconfdir=/etc
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
